STATISTICAL RESEARCH REPORT No 10
Institute of Mathematics October 1977
University of Oslo

A COMPUTATIONAL ROUTINE FOR THE CUMULATIVE ECCENTRIC FISHER DISTRIBUTION

by

Helge-Ivar Magnussen
and
Bjørn Sundt

INSTITUTE OF MATHEMATICS
University of Oslo

April 1979

Correction to

A COMPUTATIONAL ROUTINE FOR THE CUMULATIVE ECCENTRIC FISHER DISTRIBUTION (Statistical Research Report No. 10, 1977).

A disadvantage of the program in section 4 of the Research Report, is that routine 4 on p. 12 gives "Error" display when $\lambda = 0$. The following modified version of the program avoids this disadvantage. The program description on pp. 12-13 is still valid.

Bjørn Sundt

Program on HP 67 calculator for $G_{\mu,\nu}f;\mu$ when not both μ and ν are odd.

```
f LBL A
h x≷y
h CF 0
2
÷
STO A
g FRAC
f x=0
h SF 0
h x≷y
2
÷
STO B
h F? 0
h RTN
g FRAC
h N!
h RTN
f LBL B
STO C
h x≷y
RCL A
h ST I
×
STO 6
RCL B
+
STO÷6
1
STO 4
STO 7
RCL 6
-
STO D
RCL B
STO 5
h y^x
h x≷y
h F? 0
GTO 3
RCL A
1
+
STO 5
RCL 6
RCL A
h y^x
STO 4
RCL D
×
RCL B
1
-
h ST I
f x=0
GTO 2
f LBL 3
h x≷y
STO-4
RCL 5
×
RCL 7
÷
h F? 0
GTO 4
STO+4
RCL D
×
GTO 5
f LBL 4
RCL 6
×
f LBL 5
1
STO+5
STO+7
f DSZ
GTO 3
f LBL 2
h x≷y
STO E
RCL C
f LBL C
f x≠0
GTO 1
RCL 4
h RTN
f LBL 1
ENTER
2
÷
STO C
CHS
g e^x
STO 2
RCL 4
STO 0
×
STO 1
RCL E
h ST I
RCL 5
STO 9
1
STO 8
RCL A
+
STO 3
f LBL 6
h RC I
STO-0
RCL 9
×
RCL 3
÷
RCL 6
×
h ST I
RCL 8
RCL C
÷
STO÷2
1
STO+3
STO+8
STO+9
-
EEX
5
÷
RCL 2
RCL 0
×
STO+1
g x>y
GTO 6
RCL 1
h RTN
```

Abstract.

A routine for computing the cumulative eccentric Fisher distribution is developed in the case when at least one of the degrees of freedom is even. Programs on the Hewlett-Packard calculators HP-25 and HP-67 are included.

1. Some relations between distribution functions.

1 A. Let $b_n(x;p)$ and $B_n(x;p)$ denote the binomial probabilities

(1) $$b_n(x;p) = \binom{n}{x}p^x(1-p)^{n-x}$$

(2) $$B_n(x;p) = \sum_{j=o}^{x} b_n(j;p)$$

with $x=0,1,\ldots,n$ and $n=0,1,2,\ldots$.

We extend these definitions in two directions:

i) $x=0,1,\ldots,[n]$; $0 \leq n < \infty$.

We still use (1) and (2) as definitions.

ii) $n-x = 0,1,2,\ldots$; $0 \leq x < \infty$.

$b_n(x;p)$ and $B_n(x;p)$ are defined by

$$b_n(x;p) = b_n(n-x;1-p)$$

$$B_n(x;p) = 1-B_n(n-x-1;1-p).$$

The following relations are easily verified:

(3) $$B_{n+1}(x;p) = B_n(x;p) - pb_n(x;p)$$

(4) $$b_{n+1}(x;p) = \frac{n+1}{n+1-x}(1-p)b_n(x;p)$$

(5) $$B_n(x+1;p) = B_n(x;p) + b_n(x+1;p)$$

(6) $$b_{n+1}(x+1;p) = \frac{n+1}{x+1}pb_n(x;p) .$$

1 B. For the Poisson probability

$$b(x;\lambda) = \frac{\lambda^x}{x!} e^{-\lambda} \quad , \qquad x=0,1,2,\ldots$$

we have

(7) $$b(x+1;\lambda) = \frac{\lambda}{x+1} b(x;\lambda).$$

1 C. Let $K_{\mu,\nu}(g)$ be the central cumulative distribution of the ratio between chi-squares with μ and ν degrees of freedom, and let

$$H_{\mu,\nu}(g) = 1 - K_{\mu,\nu}(g).$$

We have

(8) $$K_{\mu,\nu}(g) = H_{\nu,\mu}(\tfrac{1}{g})$$

(9) $$H_{\mu,\nu}(g) = \frac{\Gamma(\frac{\mu+\nu}{2})}{\Gamma(\frac{\mu}{2})\Gamma(\frac{\nu}{2})} \int_g^\infty \frac{t^{\frac{\mu}{2}-1}}{(1+t)^{\frac{\mu+\nu}{2}}} dt.$$

Substituting $z = \frac{t}{1+t}$ in (9) gives

(10) $$H_{\mu,\nu}(g) = \frac{\Gamma(\frac{\mu+\nu}{2})}{\Gamma(\frac{\mu}{2})\Gamma(\frac{\nu}{2})} \int_{\frac{g}{1+g}}^1 z^{\frac{\mu}{2}-1}(1-z)^{\frac{\nu}{2}-1} dz.$$

By partial integration we get

(11) $$H_{\mu,\nu}(g) = \frac{\Gamma(\frac{\mu+\nu}{2})}{\Gamma(\frac{\mu}{2})\Gamma(\frac{\nu}{2}+1)} \left(\frac{g}{1+g}\right)^{\frac{\mu}{2}-1} \left(\frac{1}{1+g}\right)^{\frac{\nu}{2}} + H_{\mu-2,\nu+2}(g).$$

(10) gives

(12) $$H_{2,\mu+\nu-2}(g) = \left(\frac{1}{1+g}\right)^{\frac{\mu+\nu}{2}-1} .$$

When μ is even, we get

(13) $$H_{\mu,\nu}(g) = B_{\frac{\mu+\nu}{2}-1}\left(\frac{\mu}{2}-1;\frac{g}{1+g}\right)$$

by solving the difference equation (11) with the initial condition (12).

When ν is even, we have

$$H_{\mu,\nu}(g) = K_{\nu,\mu}\left(\frac{1}{g}\right) = 1 - B_{\frac{\mu+\nu}{2}-1}\left(\frac{\nu}{2}-1;\frac{1}{1+g}\right)$$

$$= B_{\frac{\mu+\nu}{2}-1}\left(\frac{\mu}{2}-1;\frac{g}{1+g}\right),$$

so that (13) is valid in this case too.

1 D. Let $G_{\mu,\nu}(f;\lambda)$ be the cumulative eccentric Fisher distribution with μ and ν degrees of freedom and eccentricity λ. We will assume that either μ or ν is even. Then we have

$$G_{\mu,\nu}(f;\lambda) = EK_{\mu+2I,\nu}\left(f\frac{\mu}{\nu}\right),$$

where I is Poisson distributed with $EI = \frac{\lambda}{2}$. This gives

$$G_{\mu,\nu}(f;\lambda) = EK_{\mu+2I,\nu}\left(f\,\frac{\mu}{\nu}\right) =$$

$$EH_{\nu,\mu+2I}\left(\frac{\nu}{f\mu}\right) = EB_{\frac{\mu+\nu}{2}-1+I}\left(\frac{\nu}{2}-1;\frac{\nu}{\nu+f\mu}\right) =$$

$$\sum_{i=0}^{\infty} b\left(i;\frac{\lambda}{2}\right) B_{\frac{\mu+\nu}{2}-1+i}\left(\frac{\nu}{2}-1;\frac{\nu}{\nu+f\mu}\right).$$

Letting

(14) $$a_i = b\left(i;\frac{\lambda}{2}\right) B_{\frac{\mu+\nu}{2}-1+i}\left(\frac{\nu}{2}-1;\frac{\nu}{\nu+f\mu}\right) \quad , \quad i=0,1,\ldots,$$

we have

(15) $$G_{\mu,\nu}(f;\lambda) = \sum_{i=0}^{\infty} a_i \, .$$

2. How to compute $G_{\mu,\nu}(f;\lambda)$.

2 A. To compute $G_{\mu,\nu}(f;\lambda)$, we compute the a_i's successively, adding them, until they start decreasing (they may first increase) and become insignificantly small.

2 B. We suggest the following routine for computing the a_i's:



2 C. To compute $\frac{\nu}{\nu+f\mu}\ b_{\frac{\mu+\nu}{2}-1}(\frac{\nu}{2}-1;\frac{\nu}{\nu+f\mu})$ and $B_{\frac{\mu+\nu}{2}-1}(\frac{\nu}{2}-1;\frac{\nu}{\nu+f\mu})$ there are several ways to go. We suggest the following:

Case i). μ is even.

Start with

$$B_{\frac{\nu}{2}-1}(\tfrac{\nu}{2}-1;\tfrac{\nu}{\nu+f\mu}) = 1$$

$$\frac{\nu}{\nu+f\mu}\ b_{\frac{\nu}{2}-1}(\tfrac{\nu}{2}-1;\tfrac{\nu}{\nu+f\mu}) = (\tfrac{\nu}{\nu+f\mu})^{\frac{\nu}{2}} ;$$

then compute $B_{\frac{\nu}{2}-1+j}(\frac{\nu}{2}-1;\frac{\nu}{\nu+f\mu})$ and $\frac{\nu}{\nu+f\mu}\ b_{\frac{\nu}{2}-1+j}(\frac{\nu}{2}-1;\frac{\nu}{\nu+f\mu})$ recursively, using (3) and (4), until $j = \frac{\mu}{2}$.

Case ii). ν is even.

Start with

$$B_{\frac{\mu}{2}}(-1;\tfrac{\nu}{\nu+f\mu}) = 0$$

$$b_{\frac{\mu}{2}}(0;\tfrac{\nu}{\nu+f\mu}) = (\tfrac{f\mu}{\nu+f\mu})^{\frac{\mu}{2}} ;$$

then compute $B_{\frac{\mu}{2}+j}(j;\frac{\nu}{\nu+f\mu})$ and $\frac{\nu}{\nu+f\mu}\ b_{\frac{\mu}{2}+j}(j;\frac{\nu}{\nu+f\mu})$ recursively, using (3), (5), and (6), until $j = \frac{\nu}{2}-1$.

Alternatively, compute $B_{\frac{\mu+\nu}{2}-1}(\frac{\mu}{2}-1;\frac{f\mu}{f\mu+\nu})$ and $\frac{f\mu}{f\mu+\nu}\ b_{\frac{\mu+\nu}{2}-1}(\frac{\mu}{2}-1;\frac{f\mu}{f\mu+\nu})$ as described in case i), and use that

$$B_{\frac{\mu+\nu}{2}-1}(\tfrac{\nu}{2}-1;\tfrac{\nu}{\nu+f\mu}) = B_{\frac{\mu+\nu}{2}-1}(\tfrac{\mu}{2}-1;\tfrac{f\mu}{f\mu+\nu})$$

$$\frac{\nu}{\nu+f\mu}\, b_{\frac{\mu+\nu}{2}-1}\left(\frac{\nu}{2}-1;\frac{\nu}{\nu+f\mu}\right) = \frac{\frac{f\mu}{f\mu+\nu}\, b_{\frac{\mu+\nu}{2}-1}\left(\frac{\mu}{2}-1;\frac{f\mu}{f\mu+\nu}\right)}{f} .$$

2 D. The following theorem (Sverdrup (1976)) gives an upper bound for the inaccuracy we get by adding only a finite number of a_i's .

Theorem: Let $m>\frac{\lambda}{2}$. Then

$$\sum_{i=m+1}^{\infty} a_i < a_m \frac{\frac{\lambda}{2}}{m-\frac{\lambda}{2}} . \tag{16}$$

Proof: By (3) we see that $B_{n+1}(x;p)<B_n(x;p)$. We get

$$\sum_{i=m+1}^{\infty} a_i = \sum_{i=m+1}^{\infty} b(i;\tfrac{\lambda}{2})B_{\frac{\mu+\nu}{2}-1+i}\left(\frac{\nu}{2}-1;\frac{\nu}{\nu+f\mu}\right) <$$

$$B_{\frac{\mu+\nu}{2}-1+m}\left(\frac{\nu}{2}-1;\frac{\nu}{\nu+f\mu}\right) \sum_{i=m+1}^{\infty} \frac{(\frac{\lambda}{2})^i}{i!} e^{-\lambda} =$$

$$a_m \sum_{i=m+1}^{\infty} \frac{(\frac{\lambda}{2})^{i-m}}{i(i-m)} < a_m \sum_{i=m+1}^{\infty} \left(\frac{\lambda}{2m}\right)^{i-m} = a_m \frac{\frac{\lambda}{2}}{m-\frac{\lambda}{2}} .$$

Q.E.D.

Suppose that we want the inaccuracy to be less than a fixed ϵ . By the theorem, this could be achieved by adding the a_i's until

$$a_i \le \left(\frac{i}{\frac{\lambda}{2}}-1\right)\epsilon .$$

2 E. Sverdrup (1976) describes how a program for the cumulative eccentric Fisher distribution can also be used to computations on the eccentric chi-square and Student distribution. He also gives programs on HP-25.

Guenther (1975) gives a computational routine for the cumulative eccentric Fisher distribution when the cumulative central Fisher distribution is pre-programmed.

3. Programs on HP-25 calculator.

3 A. Program for $G_{\mu,\nu}(f;\lambda)$ when μ is even.

1	STO 3	17	STO-2	33	f $x \geq y$
2	CHS	18	STO 5	34	GTO 19
3	g e^x	19	RCL 0	35	RCL 4
4	STO 4	20	STO-5	36	RCL 5
5	RCL 2	21	1	37	×
6	RCL 2	22	STO+6	38	STO+7
7	RCL 1	23	RCL 2	39	f PAUSE
8	RCL 0	24	RCL 6	40	RCL 3
9	×	25	÷	41	RCL 6
10	+	26	+	42	RCL 1
11	÷	27	RCL 1	43	f INT
12	STO-1	28	g FRAC	44	-
13	RCL 2	29	×	45	÷
14	f y^x	30	STO× 0	46	STO× 4
15	STO 0	31	RCL 6	47	GTO 19
16	1	32	RCL 1		

Set: f REG, f, STO 0, $\frac{\mu}{2}$, STO 1, $\frac{\nu}{2}$, STO 2, $\frac{\lambda}{2}$, R/S .

Observe the number shown by pause until it starts decreasing and becomes insignificantly small. R/S, RCL 7 gives $G_{\mu,\nu}(f;\lambda)$.

Registers:

0 $\frac{\nu}{\nu+f\mu}\, b_{\frac{\nu}{2}-1+j}(\frac{\nu}{2}-1;\frac{\nu}{\nu+f\mu})$

1 $\frac{\mu}{2}\frac{\nu}{\nu+f\mu}$

2 $\frac{\nu}{2}-1$

3 $\frac{\lambda}{2}$

4 $b(j-\frac{\mu}{2};\frac{\lambda}{2})$

5 $B_{\frac{\nu}{2}-1+j}(\frac{\nu}{2}-1;\frac{\nu}{\nu+f\mu})$

6 j

7 $\sum_{i=o}^{j-\frac{\mu}{2}} a_i$

As the present program has a more convenient input than the program in 3 B, we recommend the present one to be used when both μ and ν are even.

For the same reason the present program is preferable in the sentral case when μ is odd and ν is even, by using the relation

$$G_{\mu,\nu}(f) = 1-G_{\nu,\mu}(\frac{1}{f}) .$$

3 B. Program for $G_{\mu,\nu}(f;\lambda)$ when ν is even.

1	STO 4	17	GTO 26	33	STO+4
2	f y^x	18	-	34	STO+1
3	RCL 2	19	÷	35	RCL 0
4	g e^x	20	1	36	RCL 4
5	÷	21	+	37	÷
6	STO+6	22	RCL 5	38	+
7	1	23	STO-6	39	RCL 3
8	STO-1	24	×	40	×
9	RCL 3	25	GTO 06	41	RCL 2
10	-	26	STO-1	42	RCL 1
11	×	27	RCL 6	43	÷
12	STO 5	28	STO+7	44	STO× 6
13	RCL 4	29	f PAUSE	45	×
14	RCL 0	30	RCL 5	46	STO× 5
15	RCL 1	31	STO-6	47	GTO 27
16	g $x<0$	32	1		

Set: f REG, $\frac{\nu}{2}-1$, STO 0, STO 1, $\frac{\lambda}{2}$, STO 2, $\frac{f\mu}{f\mu+\nu}$, STO 3, $\frac{\mu}{2}$, R/S .

Observe the number shown by pause until it starts decreasing and becomes insignificantly small. R/S, RCL 7 gives $G_{\mu,\nu}(f;\lambda)$.

Registers:

0 $\frac{\nu}{2}-1$

1 $\frac{\nu}{2}-1-j;i$

2 $\frac{\lambda}{2}$

3 $\frac{f\mu}{\nu+f\mu}$

4 $\frac{\mu}{2}+i$

5 $\frac{\nu}{\nu+f\mu}$ $b(0;\frac{\lambda}{2})$ $b_{\frac{\mu}{2}+j}(j;\frac{\nu}{\nu+f\mu})$; $\frac{\nu}{\nu+f\mu}$ $b(i;\frac{\lambda}{2})$ $b_{\frac{\mu+\nu}{2}-1+i}(\frac{\nu}{2}-1;\frac{\nu}{\nu+f\mu})$

6 $b(0;\frac{\lambda}{2})$ $B_{\frac{\mu}{2}+j}(j;\frac{\nu}{\nu+f\mu})$; a_i

7 $\sum_{k=0}^{i} a_k$

3 C. Example. $\mu = 4$, $\nu = 6$, $f = 4.5337$, $\lambda = 4$.
The following numbers are observed by pause:

0.12857, 0.24305, 0.22541, 0.13701, 0.06153, 0.02182,
0.00637, 0.00158, 0.00034, 0.00006, 0.00001, $1.67\cdot 10^{-6}$.

R/S, RCL 7 gives $G_{\mu,\nu}(f;\lambda) = 0.82576$. By (16) the inaccuracy is less than $\frac{2}{11-2}\cdot 1.67\cdot 10^{-6} = 3.70\cdot 10^{-7}$.

4. Program on HP-67 calculator for $G_{\mu,\nu}(f;\mu)$ when not both μ and ν are odd.

```
  1  f LBL A        36  STO 5          71  RCL 6         106  +
  2  h x≷y          37  h y^x          72  ×             107  STO 3
  3  h CF 0         38  h x≶y          73  f LBL 5       108  f LBL 6
  4  2              39  h F? 0         74  1             109  h RCL I
  5  ÷              40  GTO 3          75  STO+5         110  STO-0
  6  STO A          41  RCL A          76  STO+7         111  RCL 9
  7  g FRAC         42  1              77  f DSZ         112  ×
  8  f x=0          43  +              78  GTO 3         113  RCL 3
  9  h SF 0         44  STO 5          79  f LBL 2       114  ÷
 10  h x≷y          45  RCL 6          80  h x≶y         115  RCL 6
 11  2              46  RCL A          81  STO E         116  ×
 12  ÷              47  h y^x          82  RCL C         117  h STO I
 13  STO B          48  STO 4          83  f x≠0         118  RCL 8
 14  h F? 0         49  RCL D          84  GTO C         119  RCL C
 15  h RTN          50  ×              85  RCL 4         120  ÷
 16  g FRAC         51  RCL B          86  h RTN         121  STO÷2
 17  h N!           52  1              87  f LBL C       122  1
 18  h RTN          53  -              88  ENTER         123  STO+3
 19  f LBL B        54  h STO I        89  2             124  STO+8
 20  STO C          55  f x=0          90  ÷             125  STO+9
 21  h x≶y          56  GTO 2          91  STO C         126  -
 22  RCL A          57  f LBL 3        92  CHS           127  EEX
 23  h STO I        58  h x≶y          93  g e^x         128  5
 24  ×              59  STO-4          94  STO 2         129  ÷
 25  STO 6          60  RCL 5          95  RCL 4         130  RCL 2
 26  RCL B          61  ×              96  STO 0         131  RCL 0
 27  +              62  RCL 7          97  ×             132  ×
 28  STO÷6          63  ÷              98  STO 1         133  STO+1
 29  1              64  h F? 0         99  RCL E         134  g x>y
 30  STO 4          65  GTO 4         100  h STO I       135  GTO 6
 31  STO 7          66  STO+4         101  RCL 5         136  RCL 1
 32  RCL 6          67  RCL D         102  STO 9         137  h RTN
 33  -              68  ×             103  1
 34  STO D          69  GTO 5         104  STO 8
 35  RCL B          70  f LBL 4       105  RCL A
```

This program may be used if either μ or ν is even.
The program is normally recorded on a magnetic card. It is executed by the following routine:

1. Enter both sides of the magnetic card.
2. Set: μ , ENTER, ν , A . If none of them are even, ERROR will appear.
3. Set: f , ENTER, λ , B .
   $F_{\mu,\nu}(f;\lambda)$ is then calculated. This step may be repeated for new values of f and λ .
4. To repeat the calculations for new values of λ only, just enter the new value of λ at C . This saves especially much time because it is not necessary to calculate new **values** of the central fisher distribution.

The calculations are interuptet when $a_i < (\frac{\lambda}{2}^{i} - 1)\varepsilon$. The value of ε is set to 10^{-5} which should give at least 4 significant decimals, but this may be changed to 10^{-n} by replacing 5 step 128 by n .

The following registers are used:

A $\frac{\mu}{2}$

B $\frac{\nu}{2}$

C $\lambda;\frac{\lambda}{2}$

D $\frac{\nu}{\nu+f\mu}$

E $\frac{\nu}{\nu+f\mu}\ b_{\frac{\mu}{2}+1+i}\ (i;\frac{\nu}{\nu+f\mu});\frac{\nu}{\nu+f\mu}\ b_{\frac{\nu}{2}+i}\ (\frac{\nu}{2}-1;\frac{\nu}{\nu+f\mu})$

I $\frac{\nu}{\nu+f\mu}\ b_{\frac{\mu+\nu}{2}+j}\ (j;\frac{\nu}{\nu+f\mu})$

0 $B_{\frac{\nu+\mu}{2}+j}(j;\frac{\nu}{\nu+f\mu})$

1 $F_{\mu,\nu}(f;\lambda)$

2 $b(j;\frac{\lambda}{2})$

3 $\frac{\mu}{2}+1+j$

4 $B_{\frac{\mu}{2}+1+i}(i;\frac{\nu}{\nu+f\mu})$; $B_{\frac{\nu}{2}+i}(\frac{\nu}{2}-1;\frac{\nu}{\nu+f\mu})$

5 $\frac{\mu}{2}+1+i$; $\frac{\nu}{2}+i$

6 $\frac{f\mu}{\nu+f\mu}$

7 $i+1$

8 $j+1$

9 $\frac{\nu+\mu}{2}+j$

Register C,0,1,2,3,7,8,9 and the protected registers may be used freely without destroying the intermediat answers, but the others may not.

Acknowledgement.

We thank Professor Erling Sverdrup for having gone through the manuscript.

References.

Guenther, W.C. (1975): Evaluation of noncentral distribution integrals with a desk calculator. Research paper. College of Commerce and Industry. University of Wyoming.

Sverdrup, E. (1976): Programs on HP-25 desk calculator of mathematical functions for use in statistics. Statistical research report. Institute of mathematics. University of Oslo.